Hironaka’s Example: Finding the normalized dilatations of filled subspaces

Hoping to find out whether there are finitely many or infinitely many subspaces with filled normalized dilatations less than the minimizer for Hironaka’s 3-manifold.

Set up functions

First we set up function that, given a primitive integral class (a, b) in the fibered cone in Hironaka’s example, compute the genus $g_{(a,b)}$, the number of boundary components $\partial_{(a,b)}$, and the dilatation $\lambda(a, b)$ as per Hironaka Proposition 3.4, Proposition 3.3, and Proposition 3.1 respectively.

In[1]:=
```
genus[a_, b_] := Module[{genus},
                                   If[GCD[3, a * b] == 1, genus = Abs[b], genus = Abs[b] - 1];
                                   genus]
boundary[a_, b_] := Module[{boundary},
                                   If[GCD[3, a * b] == 1, boundary = 2, boundary = 4];
                                   boundary]
dila[a_, b_] := Module[{theta},
                         theta[t_] = Power[t, 2 * b] - t^b (1 + t^a + Power[t, -a]) + 1;
                         Max[Abs[t /. Solve[theta[t] == 0, t, Reals]]]]
```

Normalized dilatation functions

Next we compute the normalized dilatation $\overline{\lambda}(a, b)$ of an unfilled class primitive class (a, b) in three different ways.

The first function does this first by computing the dilatation $\lambda_{(a,b)}$ as the largest real root (in absolute value) of the corresponding Teichmuller polynomial: $t^{2b} - t^b\left(1 + t^a + t^{-a}\right) + 1$ (Hironaka Proposition 3.1).

Then it computes the Thurston norm: $\| (a, b) \|_T = 2b$ (Hironaka Equation 2).

Finally, the normalized dilatation is $\| (a, b) \|_T \log\left(\lambda_{(a,b)}\right)$.

In[4]:=
```
normDilaUnfilled1[a_, b_] := 2 * b * Log[dila[a, b]]
```

The second function accomplishes the same task in a very similar manner, however instead of using the Thurston norm directly, it is computed indirectly by calculating the genus ($g_{(a,b)} = b$ if $\gcd(3, a\,b) = 1$ and $g_{(a,b)} = b - 1$ if $\gcd(3, a\,b) = 3$ (Hironaka Proposition 3.4)) and the number of boundary components ($\partial_{(a,b)} = 2$ if $\gcd(3, a\,b) = 1$ and $\partial_{(a,b)} = 4$ if $\gcd(3, a\,b) = 3$ (Hironaka Proposition 3.3)). The Thurston norm is then $\| (a, b) \|_T = 2\, g_{(a,b)} + \partial_{(a,b)} - 2$.

This version is less desirable than the first because it requires more calculations and it can only be carried out with integer classes due to the gcd computations.

In[5]:=
```
normDilaUnfilled2[a_, b_] := (2 * genus[a, b] + boundary[a, b] - 2) * Log[dila[a, b]]
```

The third function takes a slightly different approach.
We use the fact that the normalized dilatation is constant on rays through the origin (Hironaka Corollary 2.3).
First, we find the level set where $\log\left(\lambda_{(a,b)}\right) = 1$ by solving for when $\lambda_{(a,b)} = e$ which is equivalent to solving the equation $e^{2\,b} - e^{b}\left(1 + e^{a} + e^{-a}\right) + 1 = 0$ and writing this as a function of a:

In[6]:=
```
levelset[a_] := b /. Solve[Exp[2 b] - Exp[b] * (1 + Exp[a] + Exp[-a]) + 1 == 0, b, Reals][[2]]
```

We graph this level set below together with the boundary of the fibered cone in question.

In[7]:=
```
p1 = Plot[levelset[a], {a, -10, 10}, PlotStyle → Red];
p2 = Plot[Abs[a], {a, -10, 10}];
Show[p1, p2]
```

Solve : Solve was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result.

Out[9]=



On this level set, the normalized dilatation just the Thurston norm.
Therefore, to compute the normalized dilatation for a class (a, b) we must find the intersection point of the ray through $(0, 0)$ and (a, b) and the above level set.
The normalized dilatation is the the Thurston norm of the (y value of) this intersection point.
(This seems to be the most computationally expensive method).

In[10]:=
```
normDilaUnfilled3[a_, b_] := Module[{xval},

  xval = x /. Solve[levelset[x] == (b / a) * x, x][[1]];
                                         2 * levelset[xval]]
```

Now a quick check to make sure all three functions give us the same output at some test values:

In[11]:=
```
N[normDilaUnfilled1[1, 3]] == N[normDilaUnfilled2[1, 3]] == N[normDilaUnfilled3[1, 3]]
N[normDilaUnfilled1[5, 9]] == N[normDilaUnfilled2[5, 9]] == N[normDilaUnfilled3[5, 9]]
N[normDilaUnfilled1[3, 7]] == N[normDilaUnfilled2[3, 7]] == N[normDilaUnfilled3[3, 7]]
```

••• Solve : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[11]=

```
True
```

••• Solve : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[12]=

```
True
```

••• Solve : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[13]=

```
True
```

This next line shows the minimal normalized dilatation achieved in Hironaka's example.

In[14]:=
```
minimal = normDilaUnfilled1[0, 1]
```

Out[14]=

$$2\,\mathrm{Log}\left[\frac{1}{2}\left(3+\sqrt{5}\right)\right]$$

Next we turn to computing the normalized dilatations of the filled in primitive classes $(a, \hat{b})$.
The first method computed the dilatation as before, and then computes the genus as per Hironaka Proposition 3.4.

Then the normalized dilatation of $(a, \hat{b})$ is $\left(2\, g_{(a,b)} - 2\right) \lambda\,(a, b)$.

In[15]:=
```
normDilaFilled1[a_, b_] := (2 * genus[a, b] - 2) * Log[dila[a, b]]
```

The second method, instead of computing the genus directly, uses the Thurston norm of the unfilled class and subtracts the number of boundary components as per Hironaka Proposition 3.3.

In[16]:=
```
normDilaFilled2[a_, b_] := (2 * b - boundary[a, b]) * Log[dila[a, b]]
```

And finally we have another function to compute the normalized dilatation of the filled in class by using any one of the three filled functions from before, dividing by the old Thurston norm and multiplying by the new one:

In[17]:=
```
normDilaFilled3[a_, b_] :=
 normDilaUnfilled1[a, b] * (2 * genus[a, b] - 2) / (2 * genus[a, b] + boundary[a, b] - 2)
```

In[18]:=
```
normDilaFilled1[1, 3] == normDilaFilled2[1, 3] == normDilaFilled3[1, 3]
normDilaFilled1[5, 9] == normDilaFilled2[5, 9] == normDilaFilled3[5, 9]
normDilaFilled1[3, 7] == normDilaFilled2[3, 7] == normDilaFilled3[3, 7]
```

Out[18]=
```
True
```

Out[19]=
```
True
```

Out[20]=
```
True
```

Evidence for infinitely many filled in classes less than minimizer

Next we show evidence that there are infinitely many codimension one subspaces (lines) such that the filled in primitive class on that line has normalized dilatation less than Hironaka's minimizer.
This corresponds to the idea that there are infinitely many subspaces whose filled in minimizer is smaller than the minimizer on the unfilled manifold.
In particular, the following plots show for $b \equiv 0 \pmod 3$ with for $3 \leq b \leq 300\,000$ that
$\overline{\lambda}((1, \hat{b})) \leq 2 \log\left(\frac{3+\sqrt{5}}{2}\right)$.

In[22]:=
```
Timing[DiscretePlot[normDilaFilled1[1, b] - minimal, {b, 3, 300, 3}]]
```

Out[22]=

{215.67, 

}

In[23]:=
```
Timing[DiscretePlot[normDilaFilled1[1, b] - minimal, {b, 3, 300 000, 3}]]
```

Out[23]=

{1443.15,



}

The proof for such a claim would probably go something like as follows:

When $b \equiv 0 \pmod 3$, we get that gcd $(3, 1 \cdot b) = 3$ so that $g_{(1,b)} = b - 1$. Therefore

$\| (1, \hat{b}) \|_T = 2(b-1) - 2 = 2(b-2)$.

Thus, $\overline{\lambda}((1, \hat{b})) = 2(b-2)\log(\lambda(1, b))$. So $\overline{\lambda}((1, \hat{b})) = 2(b-2)\log(\lambda(1, b)) < 2\log\left(\frac{3+\sqrt{5}}{2}\right)$ is equivalent to $\lambda(1, b) < \frac{3+\sqrt{5}}{2}$.

The issue is that it is very difficult to come up with an analytical expression for $\lambda(1, b)$:

In[24]:=
```
dila[1, b]
```

Solve : This system cannot be solved with the methods available to Solve.

ReplaceAll : $\{\text{Solve}[1 + t^{2b} - t^b (1 + \frac{1}{t} + t) == 0, t, \mathbb{R}]\}$ is neither a list of replacement rules nor a valid dispatch table, and so cannot be used for replacing.

Out[24]=

$\text{Abs}[t \mathbin{/.} \text{Solve}[1 + t^{2b} - t^b (1 + \frac{1}{t} + t) == 0, t, \mathbb{R}]]$

The following were just tests to see which normalized dilatation computation is fastest. I guess the results aren't surprising...

In[28]:=
```
Timing[DiscretePlot[normDilaFilled1[1, b] - minimal, {b, 3, 300, 3}]]
```

Out[28]=

{197.822, 

}

In[29]:=
```
Timing[DiscretePlot[normDilaFilled2[1, b] - minimal, {b, 3, 300, 3}]]
```

Out[29]=

{203.631, 

}

In[30]:=
```
Timing[DiscretePlot[normDilaFilled3[1, b] - minimal, {b, 3, 300, 3}]]
```

Out[30]=

{205.112, 

}

In[33]:=
```
Limit[dila[1, b], b → Infinity]
```

Solve : This system cannot be solved with the methods available to Solve.

ReplaceAll : $\left\{\text{Solve}\left[1 + t^{2b} - t^{b}\left(1 + \frac{1}{t} + t\right) == 0, t, \mathbb{R}\right]\right\}$ is neither a list of replacement rules nor a valid dispatch table, and so cannot be used for replacing.

Out[33]=

$$\lim_{b\to\infty} \text{Abs}\left[t \,/.\, \text{Solve}\left[1 + t^{2b} - t^{b}\left(1 + \frac{1}{t} + t\right) == 0, t, \mathbb{R}\right]\right]$$

In[38]:=
```
Timing[DiscretePlot[normDilaFilled1[1, b] - minimal, {b, 2, 200}]]
```

Out[38]=

{92.853, 

In[49]:=
```
Timing[DiscretePlot[normDilaFilled1[5, b] - minimal, {b, 6, 600, 5}]]
```

Out[49]=

{465.652, 

Here is just an example to show that not every sequence of filled in classes in less than the minimizer.

In[54]:=
```
Timing[DiscretePlot[normDilaFilled1[a, a + 1] - minimal, {a, 5, 1000}]]
```

Out[54]=

{3437.79,



}